

Draft Guidance on Progesterone

This draft guidance, once finalized, will represent the Food and Drug Administration's (FDA's) current thinking on this topic. It does not create or confer any rights for or on any person and does not operate to bind FDA or the public. You can use an alternative approach if the approach satisfies the requirements of the applicable statutes and regulations. If you want to discuss an alternative approach, contact the Office of Generic Drugs.

Active ingredient: Progesterone

Form/Route: Capsule/Oral

Recommended studies: 2 studies

1. Type of study: Fasting
Design: Partial or fully replicated crossover design *in-vivo*
Strength: 200 mg
Subjects: Healthy males and postmenopausal females, general population. As many postmenopausal women as possible should be included in the study.
Additional Comments: Please measure baseline progesterone levels at -1.0, -0.5, and 0 hours before dosing. The mean of the pre-dose progesterone levels should be used for the baseline adjustment of the post-dose levels. Baseline concentrations should be determined for each dosing period, and baseline corrections should be period specific. If a negative plasma concentration value results after baseline correction, this should be set to 0 prior to calculating the baseline-corrected AUC. Please analyze the data using both uncorrected and corrected data. Applicants may consider using a reference-scaled average bioequivalence approach for progesterone. If using this approach, please provide evidence of high variability in the bioequivalence parameters of AUC and/or C_{max} (i.e., within-subject variability $\geq 30\%$). For detailed information on this approach, please refer to the published book chapter, *Davit B, Conner D. Reference-scaled average bioequivalence approach. In: Kanfer I, Shargel L, eds. Generic Drug Product Development – International Regulatory Requirements for Bioequivalence. New York, NY: Informa Healthcare, 2010: 271-272.*

2. Type of study: Fed
Design: Partial or fully replicated crossover design *in-vivo*
Strength: 200 mg
Subjects: Healthy males and postmenopausal females, general population.
Additional Comments: Please see additional comment above.

Analytes to measure (in appropriate biological fluid): Progesterone in plasma

Bioequivalence based on (90% CI): Progesterone

Waiver request of in-vivo testing: 100 mg based on (i) acceptable bioequivalence studies on the 200 mg strength, (ii) proportional similarity of the formulations across all strengths, and (iii) acceptable in vitro dissolution testing of all strengths.

Dissolution test method and sampling times:

Please note that a **Dissolution Methods Database** is available to the public at the OGD website at <http://www.accessdata.fda.gov/scripts/cder/dissolution/>. Please find the dissolution information for this product at this website. Please conduct comparative dissolution testing on 12 dosage units each of all strengths of the test and reference products. Specifications will be determined upon review of the application.

Method for Statistical Analysis Using the Reference-Scaled Average Bioequivalence Approach for Progesterone Capsules:

- Step 1.** Determine s_{WR} , the within-subject standard deviation (SD) of the reference product, for the pharmacokinetic (PK) parameters AUC and Cmax.
- If $s_{WR} < 0.294$, use the two one-sided tests procedure to determine bioequivalence (BE) for the individual PK parameter(s)
 - If $s_{WR} \geq 0.294$, use the reference-scaled procedure to determine BE for the individual PK parameter(s)

Calculation for s_{WR} can be conducted as follows:

$$s_{WR}^2 = \frac{\sum_{i=1}^m \sum_{j=1}^{n_i} (D_{ij} - \bar{D}_i)^2}{2(n - m)}$$

Where:

i = number of sequences m used in the study
[$m=3$ for partially replicated design: TRR, RTR, and RRT;
 $m=2$ for fully replicated design: TRTR and RTRT]

j = number of subjects within each sequence

T = Test product
R = Reference product

$D_{ij} = R_{ij1} - R_{ij2}$ (where 1 and 2 represent replicate reference treatments)

$$\bar{D}_i = \frac{\sum_{j=1}^{n_i} D_{ij}}{n_i}$$

$n = \sum_{i=1}^m n_i$ (i.e. total number of subjects used in the study, while n_i is number of subjects used in sequence i)

Mixed scaling: AUC (AUC_{0-t} and AUC_{0-∞}, as applicable) and Cmax may have different s_{WR} values. Only use the reference-scaled procedure for the specific PK parameter that has a $s_{WR} \geq 0.294$. The two one-sided tests procedure must be used for PK parameters with $s_{WR} < 0.294$.

Continue with steps 2 and 3 for PK parameters that have a $s_{WR} \geq 0.294$.

Continue with steps 2 and 3 for PK parameters that have a $s_{WR} \geq 0.294$.

Step 2. Determine the 95% upper confidence bound for:

$$\left(\bar{Y}_T - \bar{Y}_R \right)^2 - \theta s_{WR}^2$$

Where:

- \bar{Y}_T and \bar{Y}_R are the means of the ln-transformed PK endpoint (AUC and/or Cmax) obtained from the BE study for the test and reference products, respectively
- $\theta = \left(\frac{\ln(1.25)}{\sigma_{w0}} \right)^2$ (scaled average BE limit)
- and $\sigma_{w0} = 0.25$ (regulatory limit)

The method of obtaining the upper confidence bound is based on *Howe's Approximation I*, which is described in the following paper:

W. G. Howe (1974), Approximate Confidence Limits on the Mean of X+Y Where X and Y are Two Tabled Independent Random Variables, *Journal of the American Statistical Association*, 69 (347): 789-794.

Step 3. For the test product to be bioequivalent to the reference product, **both** of the following conditions must be satisfied for each PK parameter tested:

- a. the 95% upper confidence bound for $\left(\bar{Y}_T - \bar{Y}_R \right)^2 - \theta s_{WR}^2$ must be ≤ 0

AND

- b. the point estimate of the Test/Reference geometric mean ratio must fall within [0.80, 1.25]

If SAS[®] is used for statistical analysis*

- PROC MIXED should be used for fully replicated (4-way) BE studies
- PROC GLM should be used for partially replicated (3-way) BE studies

*not necessary to use SAS[®] if other software accomplishes same objectives

Example SAS Codes: partial reference-replicated 3-way design

For a bioequivalence study with the following sequence assignments in a partial reference-replicated 3-way crossover design:

	Period 1	Period 2	Period 3
Sequence 1	T	R	R
Sequence 2	R	T	R
Sequence 3	R	R	T

The following codes are an example of the determination of reference-scaled average bioequivalence for LAUCT.

Dataset containing TEST observations:

```
data test;
  set pk;
  if trt='T';
  latt=lauct;
run;
```

Dataset containing REFERENCE 1 observations:

```
data ref1;
  set ref;
  if (seq=1 and per=2) or (seq=2 and per=1) or (seq=3 and per=1);
  lat1r=lauct;
run;
```

Dataset containing REFERENCE 2 observations:

```
data ref2;
  set ref;
  if (seq=1 and per=3) or (seq=2 and per=3) or (seq=3 and per=2);
  lat2r=lauct;
run;
```

Define the following quantities:

T_{ij} = the observation on T for subject j within sequence i

R_{ijk} = kth observation ($k = 1$ or 2) on R for subject j within sequence i

I_{ij} = $T_{ij} - \frac{R_{ij1} + R_{ij2}}{2}$

and

D_{ij} = $R_{ij1} - R_{ij2}$

I_{ij} is the difference between a subject's (specifically subject j within sequence i) observation on T and the mean of the subject's two observations on R, while D_{ij} is the difference between a subject's two observations on R.

Determine I_{ij} and D_{ij}

```
data scavbe;
  merge test ref1 ref2;
  by seq subj;
  ilat=latt-(0.5*(lat1r+lat2r));
  dlat=lat1r-lat2r;
run;
```

Intermediate analysis - ilat

```
proc glm data=scavbe;
  class seq;
  model ilat=seq/clparm alpha=0.1;
  estimate 'average' intercept 1 seq 0.3333333333 0.3333333333 0.3333333333;
  ods output overallanova=iglm1;
  ods output Estimates=iglm2;
  ods output NObs=iglm3;
  title1 'scaled average BE';
run;
```

From the dataset IGLM2, calculate the following:

```
IGLM2:   pointest=exp(estimate);
         x=estimate**2-stderr**2;
         boundx=(max((abs(LowerCL)),(abs(UpperCL))))**2;
```

Intermediate analysis - dlat

```
proc glm data=scavbe;
  class seq;
  model dlat=seq;
  ods output overallanova=dglm1;
  ods output NObs=dglm3;
  title1 'scaled average BE';
run;
```

From the dataset DGLM1, calculate the following:

```
DGLM1:      dfd=df;
            s2wr=ms/2;
```

From the above parameters, calculate the final 95% upper confidence bound:

```
theta=((log(1.25))/0.25)**2;
y=-theta*s2wr;
boundy=y*dfd/cinv(0.95,dfd);
sWR=sqrt(s2wr);
critbound=(x+y)+sqrt(((boundx-x)**2)+((boundy-y)**2));
```

Example SAS Codes: fully replicated 4-way design

For a bioequivalence study with the following sequence assignments in a partial reference-replicated 4-way crossover design:

	Period 1	Period 2	Period 3	Period 4
Sequence 1	T	R	T	R
Sequence 2	R	T	R	T

The following codes are an example of the determination of reference-scaled average bioequivalence for LAUCT.

Dataset containing TEST 1 observations:

```
data test1;
  set test;
  if (seq=1 and per=1) or (seq=2 and per=2);
  lat1t=lauct;
run;
```

Dataset containing TEST 2 observations:

```
data test2;
  set test;
  if (seq=1 and per=3) or (seq=2 and per=4);
  lat2t=lauct;
run;
```

Dataset containing REFERENCE 1 observations:

```
data ref1;
  set ref;
  if (seq=1 and per=2) or (seq=2 and per=1);
  lat1r=lauct;
run;
```

Dataset containing REFERENCE 2 observations:

```
data ref2;
  set ref;
  if (seq=1 and per=4) or (seq=2 and per=3);
  lat2r=lauct;
run;
```

Further assume that there are no missing observations. All subjects provide two observations on T and two observations on R. The number of subjects in each sequence is n_1 and n_2 for sequences 1 and 2, respectively.

Define the following quantities:

T_{ijk} = kth observation ($k = 1$ or 2) on T for subject j within sequence i

R_{ijk} = kth observation ($k = 1$ or 2) on R for subject j within sequence i

I_{ij} = $\frac{T_{ij1} + T_{ij2}}{2} - \frac{R_{ij1} + R_{ij2}}{2}$

and

D_{ij} = $R_{ij1} - R_{ij2}$

I_{ij} is the difference between the mean of a subject's (specifically subject j within sequence i) two observations on T and the mean of the subject's two observations on R, while D_{ij} is the difference between a subject's two observations on R.

Determine I_{ij} and D_{ij}

```
data scavbe;
  merge test1 test2 ref1 ref2;
  by seq subj;
  ilat=0.5*(lat1t+lat2t-lat1r-lat2r);
  dlat=lat1r-lat2r;
run;
```

Intermediate analysis - ilat

```
proc mixed data=scavbe;
  class seq;
  model ilat =seq/ddfm=satterth;
  estimate 'average' intercept 1 seq 0.5 0.5/e cl alpha=0.1;
  ods output CovParms=iout1;
  ods output Estimates=iout2;
  ods output NObs=iout3;
  title1 'scaled average BE';
  title2 'intermediate analysis - ilat, mixed';
run;
```

From the dataset IOOUT2, calculate the following:

```
IOOUT2:  pointest=exp(estimate);
         x=estimate**2-stderr**2;
         boundx=(max((abs(lower)),(abs(upper))))**2;
```

Intermediate analysis - dlat

```
proc mixed data=scavbe;
  class seq;
  model dlat=seq/ddfm=satterth;
  estimate 'average' intercept 1 seq 0.5 0.5/e cl alpha=0.1;
  ods output CovParms=dout1;
  ods output Estimates=dout2;
  ods output NObs=dout3;
  title1 'scaled average BE';
  title2 'intermediate analysis - dlat, mixed';
run;
```

From the dataset DOUT1, calculate the following:

```
DOUT1:      s2wr=estimate/2;
```

From the dataset DOUT2, calculate the following:

```
DOUT2:      dfd=df;
```

From the above parameters, calculate the final 95% upper confidence bound:

```
theta=((log(1.25))/0.25)**2;
y=-theta*s2wr;
boundy=y*dfd/cinv(0.95,dfd);
sWR=sqrt(s2wr);
critbound=(x+y)+sqrt(((boundx-x)**2)+((boundy-y)**2))
```

For PK parameters with a $s_{WR} < 0.294$, use the unscaled average bioequivalence approach:

Calculation of unscaled 90% bioequivalence confidence intervals:

```
PROC MIXED
  data=pk;
  CLASSES SEQ SUBJ PER TRT;
  MODEL LAUCT = SEQ PER TRT/ DDFM=SATTERTH;
  RANDOM TRT/TYPE=FA0(2) SUB=SUBJ G;
  REPEATED/GRP=TRT SUB=SUBJ;
  ESTIMATE 'T vs. R' TRT 1 -1/CL ALPHA=0.1;
  ods output Estimates=unscl;
  title1 'unscaled BE 90% CI - guidance version';
  title2 'AUCT';
run;

data unscl;
  set unscl;
  unscabe_lower=exp(lower);
  unscabe_upper=exp(upper);
run;
```